

Excel- Basic Financial Functions

By Neale Blackwood

www.a4accounting.com.au

a4@iinet.net.au



Table of Contents

Introduction	3
PV Function	4
PMT Function	5
CUMIPMT Function	6
RATE Function	7
FV Function	8
NPER Function.....	9
Loan Model	10

Introduction

Excel has many functions dedicated to financial calculations. In most cases they are straightforward to use but do have a few issues you need to be aware of.

In this session I will examine and demonstrate a number of these functions. These functions are the more basic of the financial functions in Excel. They provide a good introduction to working with financial calculations. Even though these are the more basic financial functions they all require a number of arguments to be able to return a result.

Defaults

Many of these functions have three or more arguments. Some of the arguments are optional and use default values. This means the argument can be left out and the default value will be used. You need to be aware what the default is before you use it. Using the defaults can simplify the formula.

Rates

Many of the functions use a rate. This may be an interest rate or it may be a percentage that is used as a discounting factor. Some of these calculations relate to the time value of money, where a percentage rate is applied to a series of cash flows.

When working with rates it is always important to understand whether you are working with an annual rate or a monthly rate. In many cases you enter an annual rate and then divide it by 12 in the formulas to convert it to the monthly rate for the calculation.

Periods

Many of the functions require the entry of a number of periods. Again it is important to know if this is years or months.

Loan Calculator

The session will finish with a loan calculator where you can enter two out of three variables and the model will calculate the missing entry.

PV Function

Calculates the present value of a stream of regular payments of the same amount each period. Commonly used to value and compare different annuities.

Syntax

PV(Rate, Number_Periods, Payment, Future_Value, Type)

Rate is the interest rate per period - if working with monthly periods you may need to divide by 12.

Number_Periods is the number of periods for regular payments

Payment is the amount of the periodic payment. The payment is the opposite of the result. Negative payment = positive result.

Future_Value (optional) a different payment value for the last payment, if omitted assumed to be zero

Type (optional) (0 or 1) specifies when in the period the payment is made. If omitted, 0 is used and this assumes the payment is made at the end of the period (arrears). 1 means the payment is made at the beginning of the period (in advance). Ideally all the inputs will be cell inputs and the formula will refer to the cell inputs.

Examples

	A	B	C	D	E	F
2		Present Value Calculations	Input	Input	Input	Input
3		Rate % p.a.	4%	4%	4%	4%
4		Number of Periods	48	48	48	48
5		Payment	\$1,000.00	\$1,000.00	\$2,000.00	-\$2,000.00
6		Future Value	\$0.00	\$0.00	\$1,000.00	-\$1,000.00
7		Type	0	1	1	1
9		Present Value (all)	-\$44,288.83	-\$44,436.46	-\$89,725.30	\$89,725.30
10		Present Value (3 inputs only)	-\$44,288.83	-\$44,288.83	-\$88,577.67	\$88,577.67
12		Rate % p.a.	4%	4%	4%	4%
13		Number of Periods	4	4	4	4
14		Payment	\$12,000.00	\$12,000.00	\$24,000.00	-\$24,000.00
16		Present Value (3 inputs only)	-\$43,558.74	-\$43,558.74	-\$87,117.49	\$87,117.49

The formula in C9 is =PV(C3/12,C4,C5,C6,C7) copied across.

The formula in C10 is =PV(C3/12,C4,C5) copied across, using the defaults for the optional arguments.

The formula in C16 is =PV(C12,C13,C14) copied across, is for years and uses the defaults.

Note the differences between the calculated values in rows 9, 10 and 16. Row 10 uses the defaults for Future Value and Type (timing of payment).

PMT Function

Calculates the repayment amount for a loan based on inputs.

Syntax

PMT(Rate,Num_Periods,Present_Value,Future_Value,Type)

Rate is the interest rate per period. Rates are usually displayed as p.a and divided by 12 to calculate the per month interest rate

Num_Periods is the number of periods. Usually the years are input so multiply the years by 12 to convert to months.

Present_Value is the loan amount.

Future_Value (optional) is a “balloon” payment to be made at the end of the loan. If omitted zero is used.

Type (optional) 0 or 1 - specifies when in the period the payment is made. If omitted, zero is used and this assumes payments are made at the end of the period (in arrears). 1 indicates payments are made at the beginning of the period (in advance). Ideally all the inputs will be cell inputs and the formula will refer to the cell inputs.

Typically only three inputs are required. Loan amount, interest rate and number of years. The result is displayed as a negative if the loan amount is positive.

Example

The formula in B6 is

=PMT(B3/12,B4*12,B2)

Cell B8 multiplies the monthly repayment by the number of monthly periods to calculate the total amount paid.

Cell B9 deducts the loan amount from the total repayment to calculate the total interest.

	A	B
1		Input
2	Loan Amount	\$250,000.00
3	Interest Rate \$ p.a.	6.00%
4	Number of Years	25
5		
6	Monthly Repayment	-\$1,610.75
7		
8	Total Repayments	-\$483,226.05
9	Total Interest Paid	-\$233,226.05
10		

Data Tables

Excel's What-If option of Data Tables can be used with good effect with loans. This table shows repayment amounts for three loan amounts and three interest rates. I have a free What If webinar on my website that includes Data Tables.

	A	B	C	D
10				
11	Repayments			
12	-\$1,610.75	5.5%	6.0%	6.5%
13	\$200,000.00	-\$1,228.17	-\$1,288.60	-\$1,350.41
14	\$250,000.00	-\$1,535.22	-\$1,610.75	-\$1,688.02
15	\$300,000.00	-\$1,842.26	-\$1,932.90	-\$2,025.62

Loan Schedules

The other common requirement with loans is to produce a loan schedule of the repayments.

There are two important formulas on this report.

Cell B20 is

$\text{=ROUND}(\$B\$6*12,2)*(\text{A20}\leq\$B\$4)$

This calculates the annual repayments for the year, but will show zero if the Loan period is less than the year in column A. The end of the formula

$\text{*(A20}\leq\$B\$4)$

will return TRUE (1) or FALSE(0).

Multiplying by this will zero years that are not required.

The formula in D20 calculates the annual interest.

	A	B	C	D	E
17	Loan Schedule				
18	Years	Repayments	Principal	Interest	Balance
19	0				\$250,000
20	1	-\$19,329	-\$4,450	-\$14,879	\$245,550
21	2	-\$19,329	-\$4,725	-\$14,604	\$240,825
22	3	-\$19,329	-\$5,016	-\$14,313	\$235,809
23	4	-\$19,329	-\$5,325	-\$14,004	\$230,484
24	5	-\$19,329	-\$5,654	-\$13,675	\$224,830
25	6	-\$19,329	-\$6,003	-\$13,327	\$218,828
26	7	-\$19,329	-\$6,373	-\$12,956	\$212,455
27	8	-\$19,329	-\$6,766	-\$12,563	\$205,689
28	9	-\$19,329	-\$7,183	-\$12,146	\$198,506
29	10	-\$19,329	-\$7,626	-\$11,703	\$190,880
30	11	-\$19,329	-\$8,096	-\$11,233	\$182,783
31	12	-\$19,329	-\$8,596	-\$10,733	\$174,188
32	13	-\$19,329	-\$9,126	-\$10,203	\$165,062
33	14	-\$19,329	-\$9,689	-\$9,640	\$155,373
34	15	-\$19,329	-\$10,287	-\$9,043	\$145,086
35	16	-\$19,329	-\$10,921	-\$8,408	\$134,165
36	17	-\$19,329	-\$11,595	-\$7,735	\$122,571
37	18	-\$19,329	-\$12,310	-\$7,019	\$110,261
38	19	-\$19,329	-\$13,069	-\$6,260	\$97,192
39	20	-\$19,329	-\$13,875	-\$5,454	\$83,317
40	21	-\$19,329	-\$14,731	-\$4,598	\$68,586
41	22	-\$19,329	-\$15,639	-\$3,690	\$52,947
42	23	-\$19,329	-\$16,604	-\$2,725	\$36,343
43	24	-\$19,329	-\$17,628	-\$1,701	\$18,715
44	25	-\$19,329	-\$18,715	-\$614	\$0
45	26	\$0	\$0	\$0	\$0
46	27	\$0	\$0	\$0	\$0
47	28	\$0	\$0	\$0	\$0
48	29	\$0	\$0	\$0	\$0
49	30	\$0	\$0	\$0	\$0
50					
51		-\$483,226	-\$250,000	-\$233,226	

The table goes up to 30 years - the usual maximum for mortgages.

CUMIPMT Function

Cell D20 uses the CUMIPMT function, which calculates the cumulative interest paid between two periods. You provide the starting period number and the ending period number and the function calculates the cumulative interest based on the other arguments in the function.

Syntax

CUMIPMT(Rate, Num_Periods, Pres_Value, Start_Period, End_Period, Type)

Rate is the monthly interest rate

Num_Periods is the number or periods in the loan

Pres_Value is the loan amount

Start_Period is the first period to include in the calculation

End_Period is the last period to include in the calculation

Type (0 or 1) timing of payment. 1 = start of period and 0 = end of period.

The formula in D20 is

=IF(A20<=\$B\$4,CUMIPMT(\$B\$3/12,\$B\$4*12,\$B\$2,(A19*12)+1,A20*12,0),0)

The IF function handles the correct number of periods. The start and end periods are calculated based on the year in column A.

I have added a chart to show the split between Interest and Principle.

RATE Function

Calculates the interest rate based on a loan amount, number of periods and the monthly repayment.

Syntax

RATE(Num_Periodes,Repayment,Pres_Value,Future_Value,Type,Guess)

Num_Periodes is the number or periods in the loan

Repayment is the payment per period

Pres_Value is the loan amount

Future_Value (optional) is a “balloon” payment to be made at the end of the loan. If omitted zero is used.

Type (optional) is the timing of when the payment is made. 1 = start of period and 0 = end of period. Default is zero.

Guess (optional) you can suggest a possible rate, from which Excel will starting looking from.

The Repayment and the Present Value need to be opposite signs.

Examples

The formula in cell B6 is

=RATE(B3,B4,B2)

This has been copied to C6

Row 2 and 4 must have opposite signs.

	A	B	C
1		Input	Input
2	Loan Amount	\$20,000	-\$20,000
3	Number of Periods	48	48
4	Monthly Repayment	-\$1,500	\$1,500
6	Calculated Interest Rate	7.24%	7.24%

The last three arguments in the function are optional, and are frequently omitted.

FV Function

Calculates the future value of saving a regular amount each period.

Syntax

FV(Rate,Num_Periods,Payment,Present_Value,Type)

Rate the interest rate per period. You may need to divide by 12 to convert to month.

Num_Periods is the number of periods that the payment is made

Payment is the payment amount

Present_Value (optional) is a starting amount, default is zero value. Use this for an opening balance.

Type (optional) is the timing when the payment is made. 1 = start of period, 0 = end of period.
Default is zero.

The result will be the opposite sign to the Payment.

Example

The formula in B7 is

=FV(B5/12,B4,B3,B2)

Having a starting amount makes a difference to the interest earned as the values earns interest for the life of the payments.

If you zero the value in cell B2 you will see the impact on the interest earned in cell B8.

	A	B
1		Input
2	Starting Amount	\$50,000.00
3	Regular Amount	\$1,500.00
4	Number of periods	60
5	Interest rate p.a.	4%
6		
7	Future Value	-\$160,498.30
8	Interest Earned	-\$20,498.30

NPER Function

Calculates the number of periodic payments you need to reach a future value.

Syntax

NPER(Rate,Num_Periods,Payment,Present_Value,Type)

Rate the interest rate per period. You may need to divide by 12 to convert to month.

Payment is the monthly payment

Present_Value is the loan amount

Future_Value (optional) is a “balloon” payment to be made at the end of the loan. If omitted zero is used.

Type (optional) is the timing of when the payment is made. 1 = start of period, 0 = end of period. Zero is the default.

The result will be the opposite sign to the Payment.

Example

The formula in cell B6 below is

=NPER(B2/12,B4,B3)

	A	B	C	D	E
1		Input			
2	Interest rate p.a.	6%			
3	Loan Amount	\$150,000.00	Loan & Repayment need to be opposite signs		
4	Monthly payment	-\$1,500.00			
5					
6	Number of periods	138.975722			
7	Years	11.5813101			

Loan Model

The Loan_Model sheet contains 4 input cells. It assumes the interest rate (cell B2) is set and is not a variable. You can enter in two of the other three inputs to calculate the missing value below.

	A	B	C
1		Input - Fill in what you know	
2	Interest Rate p.a.	6.00%	Always required
3	Period In Years		
4	Loan Amount	\$250,000.00	
5	Monthly Payment	\$1,500.00	
6			
7	Calculated		
8	Number of Months	359.25	
9	Years	29.94	
10	Loan Amount		
11	Monthly Payment		
12	Total Repayment		

The formulas used are

Cell	Formula
B8	=IFERROR(ROUND(NPER(B2/12,-B5,B4),2),"")
B9	=IFERROR(ROUND(B8/12,2),"")
B10	=IFERROR(PV(B2/12,B3*12,B5),"")
B11	=IFERROR(ROUND(PMT(B2/12,B3*12,B4),2),"")
B12	=IFERROR(IF(B11<>0,B11*B3*12,""),IF(B5<>0,B5*B3*12,0))
C8	=IF(AND(B2<>0,B4<>0,B5<>0,B8=""),"Insufficient Monthly Payment","")

The "" in the formulas instruct Excel to display a blank cell.

The last formula handles the situation where the repayment amount entered doesn't cover the interest amount and hence will never pay off the loan amount. The formula will display the message **Insufficient Monthly Payment** if that is the case. It uses the IF function with the AND function - these are not covered in this session.

The IFERROR function has been used to avoid errors when there are missing inputs for a particular function. The IFERROR function was introduced in Excel 2007 and handles all of Excel errors the same way.

Syntax

IFERROR(Formula,Error_Action)


Formula is any function or formula that may return an error

Error_Action is whatever you want to do if an error is encountered. This can be another calculation, text to display, a zero or "" to display a blank cell as in the example above.

Hiding Zeroes

The sheet also hides zeroes. Some of the functions will display a zero if missing an input. To stop zeroes displaying you can use a worksheet setting in Excel Options. To open the dialog use Alt + F + T a pressed in sequence not held down.

Make sure you click OK after you un-tick the option "Show a zero in cells that have zero value".



Hiding Zeroes in a Range

You can use a custom format if you only need to hide zeros within a specific range. The following custom format will display the comma format for numbers with no decimal places and a blank cell for zeroes.

`#,##0;-#,##0;`

Having nothing after the second semi-colon instructs Excel to show nothing for zeroes.

This custom format also hides zeroes.

`#,###;-#,###`

